# Nested Evaluation of Polynomials

Consider the polynomial function $f(x) = \frac{4x^5}{15} + \frac{2x^4}{3} + \frac{4x^3}{3} + 2x^2 + 2x + 1$. Suppose we wish to evaluate it at $x = -\frac{73}{25}$, i.e., we want to find $f(-\frac{73}{25})$. Suppose further that we we will use 3 digit rounding. We define the function.

```
> restart;
> f:=x->(4/15)*x^5+(2/3)*x^4+(4/3)*x^3+2*x^2+2*x+1;
```

$$f := x \rightarrow \frac{4}{15}x^5 + \frac{2}{3}x^4 + \frac{4}{3}x^3 + 2x^2 + 2x + 1$$

We find the exact value.

```
> X:=-73/25;
```

$$X := -\frac{73}{25}$$

```
> f(X);
```

$$-\frac{4266454247}{146484375}$$

We round this value to 10 digits.

```
> evalf(%);
```

$$-29.12566099$$

Next we evaluate the answer using three digit rounding.

```
> Digits:=3;
```

$$Digits := 3$$

We change our argument to a three digit decimal.

```
> x:=evalf(X);
```

$$x := -2.92$$

We form our powers.

```
> x2:=x*x;x3:=x2*x;x4:=x3*x;x5:=x4*x;
```

$$x2 := 8.53$$

$$x3 := -24.9$$

$$x4 := 72.7$$

$$x5 := -212.$$

We multiply the powers by our coefficients.

```
> t5:=(4/15)*x5;t4:=(2/3)*x4;t3:=(4/3)*x3;t2:=2*x2;t1:=2*x;
```

$$t5 := -56.5$$

$$t4 := 48.5$$

$$t3 := -33.2$$

$$t2 := 17.1$$

$$t1 := -5.84$$

Now we add the terms.

```
> p:=t5+t4;p:=p+t3;p:=p+t2;p:=p+t1;p:=p+1;
```

$$p := -8.0$$

$$p := -41.2$$

$$p := -24.1$$

$$p := -29.9$$

$$p := -28.9$$

Notice that we needed to do 9 multiplications and 5 additions. Now we set Digits to 10 and find the relative error.

```
> Digits:=10;
```

$$Digits := 10$$

```
> relerror:=abs(f(X)-p)/abs(f(X));
```

$$relerror := 0.007747840987$$

Thus our approximation has 2 significant digits.

We can also use the RelativeError command from the NumericalAnalysis package to find the relative error.

```
> with(Student[NumericalAnalysis]);
```

[*AbsoluteError*, *AdamsBashforth*, *AdamsBashforthMoulton*, *AdamsMoulton*, *AdaptiveQuadrature*, *AddPoint*, *ApproximateExactUpperBound*, *ApproximateValue*, *BackSubstitution*, *BasisFunctions*, *Bisection*, *CubicSpline*, *DataPoints*, *Distance*, *DividedDifferenceTable*, *Draw*, *Euler*, *EulerTutor*, *ExactValue*, *FalsePosition*, *FixedPointIteration*, *ForwardSubstitution*, *Function*, *InitialValueProblem*, *InitialValueProblemTutor*, *Interpolant*, *InterpolantRemainderTerm*, *IsConvergent*, *IsMatrixShape*, *IterativeApproximate*, *IterativeFormula*, *IterativeFormulaTutor*, *LeadingPrincipalSubmatrix*, *LinearSolve*, *LinearSystem*, *MatrixConvergence*, *MatrixDecomposition*, *MatrixDecompositionTutor*, *ModifiedNewton*, *NevilleTable*, *Newton*, *NumberOfSignificantDigits*, *PolynomialInterpolation*, *Quadrature*, *RateOfConvergence*, *RelativeError*, *RemainderTerm*, *Roots*, *RungeKutta*, *Secant*, *SpectralRadius*, *Steffensen*, *Taylor*, *TaylorPolynomial*, *UpperBoundOfRemainderTerm*, *VectorLimit*]

```
> RelativeError(p,f(X));
```

$$0.007808338754$$

Differences from the above can be attributed to round-off schemes.

Next we write our polynomial in nested form, i.e., we write it as

$$\mathrm{f(x)} = \left(\left(\left(\left(\frac{4\,x}{15} + \frac{2}{3}\right)x + \frac{4}{3}\right)x + 2\right)x + 2\right)x + 1$$

We again evaluate using three digit rounding.

```
> Digits:=3;
```

$$Digits := 3$$

```
> p:=(4/15)*x;p:=p+2/3;p:=p*x;p:=p+4/3;p:=p*x;p:=p+2;p:=p*x;p:=p+2;
  p:=p*x;p:=p+1;
```

$$p := -0.779$$

$$p := -0.112$$
$$p := 0.327$$
$$p := 1.66$$
$$p := -4.85$$
$$p := -2.85$$
$$p := 8.32$$
$$p := 10.3$$
$$p := -30.1$$
$$p := -29.1$$

Notice that we now only needed 5 multiplications along with 5 additions. Let's check the relative error with this approach.

```
> Digits:=10;
```

$$Digits := 10$$

```
> RelativeError(p,f(X));
```

$$0.0008818209622$$

Not only is the nested approach more efficient, but our appproximation now has 3 significant digits.